

Guía 3.
Introducción a la programación -
Estructuras Iterativas

Ejercicio 6)

Se ingresa una lista de notas correspondientes a una evaluación de programación numeradas entre 0 y 10. Al finalizar se debe mostrar en pantalla:

- Cantidad de notas
- Promedio
- Cantidad de aprobados y no aprobados
- Porcentaje de alumnos con:
 - o Muy Bueno (8 o más)
 - o Bueno (6 o 7)
 - o Regular (4 o 5)
 - o Insuficiente (3 o menos)

Análisis:

Entrada	<ul style="list-style-type: none">• Lista de notas entre 0 y 10, ingresadas una por una• Cortar la lista si se ingresa un número negativo
Proceso	<ul style="list-style-type: none">• Leer notas hasta que se ingrese una nota fuera del rango
Salida	<ul style="list-style-type: none">• Cantidad total de notas.• Promedio.• Cantidad de aprobados y desaprobados.• Porcentaje de: Muy bueno, Bueno, Regulares, Insuficientes.

Estrategia:

1. Inicializar contadores.
2. Ingresar 10 notas entre (0 a 10).
3. Leer una nota.
4. Si está entre 0 a 10, sumar y clasificar.
5. Si no está en rango cortar la lista.
6. Calcular Promedio y Calcular porcentajes
7. Mostrar en pantalla lo siguiente:
 - Cantidad de aprobados y no aprobados
 - Porcentaje de alumnos con:
 - ❖ Muy Bueno (8 o más)
 - ❖ Bueno (6 o 7)
 - ❖ Regular (4 o 5)
 - ❖ Insuficiente (3 o menos)

Ambiente:

Variable	Tipo de Variable	Descripción
Nota	Entero	Nota ingresada
totalNotas	Entero	Contador total de notas
sumaNotas	Entero	Acumulador de notas para calcular promedio
aprobados	Entero	Contador de notas mayores o iguales
desaprobados	Entero	Contador de notas menores a 4
muyBuenos	Entero	Contador de nota de 8 o más
buenos	Entero	Contador de notas de 6 o 7
regulares	Entero	Contador de notas de 4 o 5
insuficiente	Entero	Contador de notas de 3 o menos
(porcentajeMB, porcentajeB, porcentajeR, porcentajeI)	Real	Porcentaje por categoría
promedio	Real	Promedio general de las notas

Pseudocódigo:

```
1  Proceso EvaluacionNotas
2  Definir nota, totalNotas, sumaNotas, aprobados, desaprobados Como Entero;
3  Definir muyBuenos, buenos, regulares, insuficientes Como Entero;
4  Definir porcentajeMB, porcentajeB, porcentajeR, porcentajeI, promedio Como Real;
5
6  totalNotas ← 0;
7  sumaNotas ← 0;
8  aprobados ← 0;
9  desaprobados ← 0;
10 muyBuenos ← 0;
11 buenos ← 0;
12 regulares ← 0;
13 insuficientes ← 0;
14
15 Escribir "Ingrese las notas entre 0 y 10. Para finalizar";
16 Leer nota;
17
18 Mientras nota ≥ 0 Y nota ≤ 10 Hacer
19     totalNotas ← totalNotas + 1;
20     sumaNotas ← sumaNotas + nota;
21     Si nota ≥ 8 Entonces
22         muyBuenos ← muyBuenos + 1;
23     Sino
24         Si nota ≥ 6 Entonces
25             buenos ← buenos + 1;
26         Sino
27             Si nota ≥ 4 Entonces
28                 regulares ← regulares + 1;
29             Sino
30                 insuficientes ← insuficientes + 1;
31             FinSi
32         FinSi
33     FinSi
34     Si nota ≥ 4 Entonces
35         aprobados ← aprobados + 1;
36     Sino
37         desaprobados ← desaprobados + 1;
38     FinSi
39     Leer nota;
40 FinMientras
41
42 Si totalNotas > 0 Entonces
43     promedio ← sumaNotas / totalNotas;
44     porcentajeMB ← (muyBuenos * 100) / totalNotas;
45     porcentajeB ← (buenos * 100) / totalNotas;
46     porcentajeR ← (regulares * 100) / totalNotas;
47     porcentajeI ← (insuficientes * 100) / totalNotas;
48
49     Escribir "cantidad total de notas: ", totalNotas;
50     Escribir "promedio general: ", promedio;
51     Escribir "cantidad de aprobados: ", aprobados;
52     Escribir "cantidad de desaprobados: ", desaprobados;
53     Escribir "porcentaje muy bueno (8 o mas): ", porcentajeMB, "%";
54     Escribir "porcentaje bueno (6 o 7): ", porcentajeB, "%";
55     Escribir "porcentaje regular (4 o 5): ", porcentajeR, "%";
56     Escribir "porcentaje insuficiente (3 o menos): ", porcentajeI, "%";
57 Sino
58     Escribir "No se ingresaron notas validas.";
59 FinSi
60 FinProceso
```